Excel to Calculate Time Value of Money

Excel allows you to easily calculate time value of money problems. There are several financial functions that work similarly to financial calculators. The functions use the same designations for all types of lump sum and annuity problems. The chart below indicates different items that are included in the functions.

Lump Sum and Annuity Problems in Excel
fv – Future Value
pv – Present Value
rate – interest rate per period (decimal form)
nper – total number of periods (number of payments for annuities)
pmt – payment per period
type – End/Begin Mode - 0 for End; 1 for Begin (You do not have to enter the type if there is no pmt.)

Notes:
1. For all financial functions in Excel, type the equal sign followed by the item you wish to find, enter the required inputs, and then press the “Enter” key to get the answer.
2. When you enter the equal sign, the item you wish to find, and then an open parenthesis,“(“, Excel will prompt you for the remaining inputs.
3. Always enter the interest rate in Excel in decimal form: 10% would be entered as 0.10.
4. If an item is not used in the problem, enter a zero.
5. When calculating the “rate” or “nper,” at least one entered cashflows must be a different sign from the other(s). Enter the outflows as negative values and the inflows as positive values.

EXAMPLES:
(A) calculate PV
Function form: =pv(rate, nper, pmt, [fv], [type])
What is the present value of $2000 received 10 years from now if the annual interest rate you could have received is 7%?
=pv(0.07, 10, 0, -2000)
Answer: $1016.70

(B) calculate FV
Function form: =fv(rate, nper, pmt, [pv], [type])
What is the future value 10 years from now of $2000 deposited today in an account which has a quoted annual interest rate of 10% with quarterly compounding of interest?
NOTE: Since the interest is compounded within the year, we have to adjust the “rate” and “nper” to reflect the quarterly payments.
=fv(0.10/4, 10*4, 0, -2000)
Answer: $5370.13

(C) calculate rate
Function Form: =rate(nper, pmt, pv, [fv], [type])
What annual interest rate must you earn if you plan to deposit $10,000 in the bank today and you want it to grow to be $17,623.42 in 5 years?
=rate(5, 0, -10000, 17623.42)
Answer: 12%
(D) calculate nper
Function Form: =nper(rate, pmt, pv, [fv], [type])
If you deposit $5000 in an account which earns 9% per year, how many years will it take for the investment to grow to be worth $10,000?
=nper(0.09, 0, -5000, 10000)
Answer: 8.0432 years

ANNUITIES
Excel allows you to solve for the various items for an ordinary annuity or an annuity due. You must make sure Excel is in the correct "mode" for the type of annuity you are considering. The “type” entry specifies whether Excel is in END or BEGIN mode for the problem. The "BEGIN" mode is used for an annuity due (begin means the payments are at the beginning of the period). The "END" mode is used for ordinary annuities (end means the payments are at the end of the period).

EXAMPLES:
(E) present value of an annuity
You are offered an investment that will pay you $5000 per year for the next 10 years. Each payment will be made at the end of each year. If the appropriate discount rate is 9% per year, what is the present value of the annuity?
=PV(0.09, 10, 5000, 0, 0)
Answer: $32,088.29

(F) future value of an annuity
You are offered an investment that will pay you $8000 per year for the next 20 years. Each payment will be made at the beginning of each year. If you expect to earn an annual return of 7% per year, what is the future value of the annuity at the end of the 20 years?
=FV(0.07, 20, 8000, 0, 1)
Answer: $350,921.41

UNEVEN CASHFLOWS
Excel will calculate the present value of an uneven stream of cash flows using the “npv” function. The Excel “npv” function is not exactly “net present value” that we will discuss later. Actually, it calculates the present value of a series of future cashflows based on the interest rate you enter.

Function Form: =npv(rate, value1, value2, …)
(The “value” entries are the cashflows. Enter them in the order that they occur.)

EXAMPLE:
Suppose that you are offered an investment which will pay $100 at the end of the first year, $200 at the end of the second year, $300 at the end of the third year, $400 at the end of the fourth year, and $500 at the end of the fifth year. How much are the cash flows worth in today’s dollars if your annual required rate of return is 12%?
=npv(0.12, 100, 200, 300, 400, 500)
Answer: $1000.18
Note: You can also enter the casflows in a series of cells and reference the cells in the formula. If yo entered the above cashflows in cells a1, a2, a3, a4, and a5, you could enter the function as:
=npv(0.12,a1:a5)